
Overview

uniquely closable skeletons of lambda terms are Motzkin-trees
that predetermine the unique closed lambda term that can be
obtained by labeling their leaves with de Bruijn indices
uniquely typable skeletons of closed lambda terms predetermine
the unique simply-typed lambda term that can be obtained by
labeling their leaves with de Bruijn indices
we derive, through a sequence of logic program transformations,
efficient code for their combinatorial generation
we obtain context-free grammars describing closable and
uniquely closable skeletons of lambda terms
⇒ we study them in depth with tools from analytic combinatorics
for the difficult case of (uniquely) typable terms: empirical study
⇒ open problems about density and asymptotic behavior
we establish a connection between the two classes of terms:
uniquely typable closed lambda term skeletons of size 3n + 1 are
in a bijection with binary trees of size n



Motivation

the study of lambda terms has theoretical ramifications:

connection to proofs in intuitionistic logic via the Curry-Howard
correspondence - the class of simply typed lambda terms

lambda terms are used in the internal representations of
compilers for functional programming languages and proof
assistants

generation of large random lambda terms helps with automated
testing

combinatorial properties need to be well understood for random
generation of large terms (e.g. via Boltzmann samplers)



Our focus:

binary-unary trees that are obtained from lambda terms in de
Bruijn form, represented as trees, by erasing the de Bruijn indices
labeling variables at their leaves

a surprising fact: such skeletons of the lambda terms turn out to
predetermine non-trivial properties the lambda terms they host,
e.g., if such terms are closed or simply-typed

⇒ what are the cases when unique such terms exist?

our declarative meta-language is Prolog

easy combinatorial generation via backtracking over the set of all
answers
a Definite Clause Grammar (DCG) enforces size constraints
we place more complex constraints at points in the code where
they ensure the earliest possible pruning of the search space
program transformations allow us to derive step-by-step faster and
simpler expressions of the underlying combinatorial mechanisms



Closed Lambda Terms and their Motzkin-trees Skeletons

a Motzkin tree (also called binary-unary tree) is a rooted ordered
tree built from binary nodes, unary nodes and leaf nodes

the set of Motzkin trees can be seen as the free algebra
generated by the constructors v/0, l/1 and a/2

lambda terms in de Bruijn form: the free algebra generated by the
constructors l/1, and a/2 and leaves labeled with natural
numbers wrapped with the constructor v/1

lambda term in de Bruijn form is closed if for each of its de Bruijn
indices it exists a lambda binder to which it points, on the path to
the root of the tree representing the term

they are counted by sequence A135501 in OEIS

skeleton of a lambda term: the Motzkin tree obtained by erasing
the labels at its leaves



Closable and Unclosable Skeletons

we call a Motzkin tree closable if it is the skeleton of at least one
closed lambda term
the predicate isClosable/1 tests if it exists a closed lambda
term having X as its skeleton
for each lambda binder it increments a count V (starting at 0),
and ensures that it is strictly positive for all leaf nodes

isClosable(X):-isClosable(X,0).

isClosable(v,V):-V>0.
isClosable(l(A),V):-succ(V,NewV),isClosable(A,NewV).
isClosable(a(A,B),V):-isClosable(A,V),isClosable(B,V).

generators for closable and unclosable skeletons: by filtering the
answers of a Motzkin tree generator motSkel/2

closableSkel(N,X):-motSkel(N,X),isClosable(X).
unClosableSkel(N,X):-motSkel(N,X),not(isClosable(X)).



Some Closable and Unclosable Motzkin Skeletons

a) 3 closable skeletons b) 3 unclosable skeletons

Figure: Closable vs. unclosable skeletons of size 7



Size definition: each constructor costs as much as its arity

Proposition

The set of terms of size n for the size definition {application=2,
lambda=1, variable=0} is equal to the set of terms of size n + 1 for the
size definition {application=1, lambda=1, variable=1}.

true, given that the number of leaves in a Motzkin tree is the
number of binary nodes + 1
the term l(a(v (0),v (0))) will have size 3 = 1 + 2 with our definition,
which corresponds to size 4 = 1 + 1 + 1 + 1 using the size definition
for sequence A135501
our size definition is implemented as

l(SX,X):-succ(X,SX). % for l/1 constructors
a-->l,l. % for a/2 constructors

with Prolog’s DCG notation controlling the consumption of size
units from N to 0



Deriving a CF Grammar

Proposition

A Motzkin tree is a skeleton of a closed lambda term if and only if it
exists at least one lambda binder on each path from the leaf to the
root.

we can derive a grammar generating closable skeletons

at least one lambda (l/1 constructor) on each path

⇒ Motzkin trees below the l/1 constructor contain only a/2
branches and leaves

this also runs about 3 times as fast as closableSkel/2

closable(N,X):-closable(X,N,0).

closable(l(Z))-->l,motSkel(Z). % a/2 and leaves only
closable(a(X,Y))-->a,closable(X),closable(Y).



Analytic Combinatorics⇒ Density of Closable Skeletons

M(z) = ∑mnzn the ordinary generating function for Motzkin trees
(mn is the number of Motzkin trees of size n)
M(z) follows the algebraic functional equation M = z + zM + zM2

M(z) = 1−z−
√
−3z2−2z+1
2z

classical result: asymptotically, the number mn of Motzkin trees of

size n is equivalent to

from our grammar definition: the ordinary generating function
C(z) for closable lambda terms is C(z) = zC(z)2 + zM(z).

consequently, C(z) = 1−
√

2z

⇒ when n tends to the infinity, the proportion of closable lambda term

skeletons tends to
1√
5
.= 44.7%.



An Efficient Recurrence Relation

we can calculate very efficiently the coefficients cn

from the C(z) = zC(z)2 + zM(z), it follows that C(z) satisfies
z2C(z)4−2zC(z)3 + (−z2 + z + 1)C(z)2 + (z−1)C(z) + z2

we derive, after several computation steps:

(
1200n5 + 18480n4 + 90816n3 + 161088n2 + 87552n

)
cn +(

800n5 + 13520n4 + 79024n3 + 202312n2 + 231768n + 95760
)

cn+1 +(
−100n5−1840n4−12848n3−38792n2−44100n−9576

)
cn+2 +(

−100n5−1990n4−14648n3−48254n2−66276n−23940
)

cn+3 +(
−225n5−4815n4−38883n3−147519n2−260286n−167580

)
cn+4 +(

150n5 + 3435n4 + 29817n3 + 120441n2 + 218739n + 131670
)

cn+5 +(
−25n5−610n4−5642n3−24128n2−45405n−26334

)
cn+6 = 0

with the initial conditions c0 = 0,c1 = 0,c2 = 1,c3 = 1,c4 = 2,c5 = 5



Uniquely Closable Skeletons

We call a skeleton uniquely closable if it exists exactly one closed
lambda term having it as its skeleton.

Proposition

A skeleton is uniquely closable if and only if exactly one lambda binder
is available above each of its leaf nodes.

Proof.

Note that if more than one were available for any leaf v, one could
choose more then one de Bruijn index at the corresponding leaf v/1
of a lambda term, resulting in more than one possible lambda terms
having the given skeleton.



A Grammar for Uniquely Closable Skeletons

By specializing with respect to having or not having a lambda binder
above, we obtain the predicate uniquelyClosable/2 which
mimics a context-free grammar generating all uniquely closable
skeletons of a given size.

uniquelyClosable(N,X):-uniquelyClosable(X,N,0).

uniquelyClosable(l(A))-->l,closedAbove(A).
uniquelyClosable(a(A,B))-->a,
uniquelyClosable(A),
uniquelyClosable(B).

closedAbove(v)-->[].
closedAbove(a(A,B))-->a,closedAbove(A),closedAbove(B).

Our program transformations result in code running an order of
magnitude faster than the original specification, with all counts up to
size 30 obtained in less than a minute.



Growth of Counts of Uniquely Closable Skeletons

Figure: Uniquely closable skeletons by increasing sizes



The Asymptotics of Uniquely Closable Skeletons

B(z) : the ordinary generating function for binary trees

the series B(z) follows the algebraic functional equation
B = z + zM2 and consequently B(z) = 1−

√
−4z2+1
2z .

the ordinary generating function U(z) for uniquely closable
lambda terms satisfies U(z) = zU(z)2 + zB(z).

consequently, U(z) = 1−
√

2z
√
−4z2+1−2z+1
2z .

we use the Flajolet-Odlysko transfer theorems

the asymptotics of the number un of uniquely closable skeletons:
un ∼ 21/4+n

4Gamma(3/4)n5/4 .



An Efficient Recurrence Relation

like for C(z) we can calculate quickly the coefficients un

U(z): z2U(z)4−2zU(z)3 + (z + 1)U(z)2−U(z) + z2 = 0
we deduce a linear differential equation:

0 =−128z5−40z4 + 52z3 + 18z2−6z + (16z5 + 56z4−20z3−20z2 + 8z−2)U(z) +

(512z8−512z7−320z6 + 96z5 + 144z4 + 16z3−24z2−6z + 2)( d
dz U (z)) +

(256z9−128z8−128z7−32z6 + 64z5 + 24z4−16z3−2z2 + z)( d2

dz2 U (z))

with the initial condition U (0) = 0.

we efficiently compute un using the P-recurrence:(
256n2 + 256n

)
un +

(
−128n2−640n−512

)
un+1 +(

−128n2−704n−880
)

un+2 +
(
−32n2−64n + 152

)
un+3 +(

64n2 + 592n + 1324
)

un+4 +
(

24n2 + 232n + 540
)

un+5 +(
−16n2−200n−616

)
un+6 +(

−2n2−32n−128
)

un+7 +
(

n2 + 17n + 72
)

un+8 = 0

with the initial conditions u0 = 0,u1 = 0,u2 = 1,u3 = 0,u4 = 1,u5 = 1,u6 = 2,u7 = 2



Typable and Untypable Closable Skeletons

we call a Motzkin skeleton typable if it exists at least one
simply-typed closed lambda term having it as its skeleton

an untypable skeleton is a closable skeleton for which no such
term exists
we design efficient the Prolog code by interleaving:

term generation,
checking for closedness
type inference steps, via unification with occurs check

we design a two stage program:
the first stage generates code
the second, executes it, via Prolog’s metacall
the second stage also ensures that the terms generated are
closed



The Two Stage Code Interleaving Generation and Type
Inference

genSkelEqs/4 generates type unification equations
if satisfied by a closed lambda term, they ensure that the term is
simply-typable

genSkelEqs(N,X,T,Eqs):-genSkelEqs(X,T,[],Eqs,true,N,0).

genSkelEqs(v,V,Vs,(el(V,Vs),Es),Es)-->{Vs=[_|_]}.
genSkelEqs(l(A),(S->T),Vs,Es1,Es2)-->l,genSkelEqs(A,T,[S|Vs],Es1,Es2).
genSkelEqs(a(A,B),T,Vs,Es1,Es3)-->a,genSkelEqs(A,(S->T),Vs,Es1,Es2),
genSkelEqs(B,S,Vs,Es2,Es3).

el(V,Vs):-member(V0,Vs),unify_with_occurs_check(V0,V).

each lambda binder adds a new type variable to the list (starting
empty at the root) on the way down to a leaf node
a term is then closed if the list of those variables Vs is not empty
at each leaf node



The Code, continued

to generate the typable terms, one simply executes the equations
Eqs:

typableClosedTerm(N,Term):-genSkelEqs(N,Term,_,Eqs),Eqs.

typableSkel/2 generates skeletons that are typable by
running the same equations Eqs and ensuring they have at least
one solution, using the Prolog built-in once/1

untypableSkel/2 succeeds, when the negation of these
equations succeeds, indicating that no simply-typed lambda term
exists having the given skeleton

this is much faster than naively generating all the closed lambda
terms and then finding their distinct skeletons

typableSkel(N,Skel):-genSkelEqs(N,Skel,_,Eqs),once(Eqs).
untypableSkel(N,Skel):-genSkelEqs(N,Skel,_,Eqs),not(Eqs).



Examples of Typeable and Untypable Skeletons

In Fig. 3 we show 3 typable and 3 untypable Motzkin skeletons.

a)
typable Motzkin skeletons b) untypable Motzkin skeletons

Figure: Typable vs. untypable skeletons of size 8



An open problem: what is the exact asymptotic behavior in
this case?

Fig. 4 compares the growth of typable and untypable skeletons.

Figure: Typable skeletons vs. untypable skeletons by increasing sizes



Uniquely Typable Skeletons and their Relation to Uniquely
Closable Skeletons

A uniquely typable skeleton is one for which it exists exactly one
simply-typed closed lambda term having it as a skeleton.



Open problem: their asymptotic behavior

Figure: Uniquely typable skeletons by increasing sizes



An Interesting Easy Case

Proposition

Uniquely closable typable skeletons of size 3n + 1 are in bijection with
Catalan objects (binary trees) of size n.

Proof.

We will exhibit a simple bijection to binary trees. We want to show that
terminal subtrees must be of the form l(v(0)). As there’s a unique
lambda above each leaf, closing it, the leaf should be (in de Bruijn
notation), v(0) pointing to the first and only lambda above it. Assume a
terminal node of the form a(v(0),v(0)). Then the two leaves must share
a lambda binder resulting in a circular term when unifying their types
(i.e., as in the case of the well-known term ω=l(a(v(0),v(0)))) and thus
it could not be typable.



Example of the Bijection to Binary Trees

two trees illustrate the shape of the skeletons and their bijection
to binary trees
the skeleton is mapped into a binary tree simply by replacing its
terminal subtrees of the form l(v) with a leaf node v

In this case, terms of size 3n + 1 = 7 = 2 + 2 + 1 + 1 + 1 + 0 + 0 + 0 are
mapped to binary trees of size n = 2 = 1 + 1 + 0 + 0 + 0 (with a/2
nodes there counted as 1 and v/0 nodes as 0) after replacing l(v)
nodes with v nodes.



No Interesting Uniquely Closable Terms that are Typable

As a consequence, each uniquely closable term that is typable is
uniquely typable, as identity functions of the form l(v(0))
would correspond to the end of each path from the root to a leaf
in a lambda term having this skeleton.

This tells us that there no “interesting” uniquely closable terms
that are typable.

However, as there are normalizable terms that are not simply
typed, an interesting open problem is to find out if closable terms,
other than those ending with l(v(0)) are (weakly)
normalizable.



Conclusions

the problem: can the shape of the underlying binary-unary tree
predetermine essential properties of lambda terms?
positive answer with exact asymptotics for the case of closed
terms
positive answer but exact asymptotics still an open problem for
simply typed terms
when simple CF grammars can be derived from the definitions of
our term classes, powerful tools from analytic combinatorics
become available
the tools used: a language as simple as (mostly) Horn Clause
Prolog can handle elegantly combinatorial generation problems
when the synergy between sound unification, backtracking and
DCGs is put at work!

the paper is organized as a literate Prolog program - our code is available at:
http://www.cse.unt.edu/~tarau/research/2017/uct.pro

http://www.cse.unt.edu/~tarau/research/2017/uct.pro


Questions?


